

## 頻度解析結果から被害度を求める

頻度解析結果から、定義した S-N 曲線を使用して被害推定演算を行う方法について説明します。  
 尚以下の説明では、頻度解析結果は、\$5 にセル毎の計数値 n が、\$6 にセルの中央値 s が格納され、セル並びは、セル中央値の昇順であることを前提としています。又、頻度解析結果のセル中央値は、全て正数である処理を前提としています。例えば、極大値・極小値解析等で正負の領域となる場合は、折り返して全て正数とします。  
 ※ セル中央値の求め方及び正負頻度解析結果の絶対値化は「被害度演算の事前処理」の項で記述します。

### <S-N 曲線をテーブルで定義した場合>

テーブルの定義は、事前に text ファイルを作成し実行時に読み込んで行う方法と、Script 記述内でテーブルを定義する方法があります。

case 1 : S-N 曲線ファイルを読み出す方法

項目区切り半角カンマで拡張子.csv のテキスト形式ファイルで与えられるものとします。尚、S-N 曲線定義ファイルは 3 行目からテーブル定義され、1 列目が S 列、2 列目が N 列と記述されているとします。

step1 : S-N 曲線テーブルファイルを読み出す

```
get file_name &1 $1 ~csv" /*ファイル読み出しダイアログを表示させ、S-N 曲線ファイル名を取得*/
&1 に選択されたファイル名、$1 に選択されたファイル個数が戻ります。
def file_id %1 &1 csv /*ファイル名&1 で取扱属性項目区切り文字半角カンマで ID を定義*/
```

step2 : テーブル値を読み出す

```
read cell %1 3,1 1 $2,$3 /*3 行 1 列目から縦方向にテーブル値を読み出す*/
※ $2 に 1 列目の S 値が、$3 に 2 列目の N 値が格納されます。
```

step3 : 読み出したテーブル値を S 値の昇順に並び替える

```
$4 = SRT(-1,$2) /*$4 に S 値の昇順キーの取得*/
$2 = QUE($4,$2) /*$4 の S 値の昇順キーにより$2 の S 値を並び替える*/
$3 = QUE($4,$3) /*$4 の S 値の昇順キーにより$3 の N 値を並び替える*/
※ テーブルファイルの記述順が順不同であることを前提とした処理です。ファイルに記載されているデータ並びが S 値の昇順で記述されている場合は、当該処理は不要です。
```

case2 : S-N 曲線テーブルを Script 記述で定義する場合

テーブルの定義は代入文 (assign 文) を使用して、S 値、及び対応した N 値をそれぞれ定義します。

step1 : テーブルの定義

```
assign $2 "S 値:" = 10,20,50,100,200,500,1000,2000,5000 /*S 値の定義*/
assign $3 "N 値:" = 5000,2000,1000,500,200,100,50,20,10 /*N 値の定義*/
※ S 値の並びは、昇順で値は唯一無二の必要があります。N 値は S 値に対応して記述し、定義する数は同じ必要があります。尚、定義例での数値は、適当な値で記述しています。
```

### <S-N 曲線をテーブルで定義した場合の被害度演算>

step1 : 被害度演算をマイナー則で行う場合の事前処理

マイナー則では s 値の指定した下限以下は被害度演算に含まない処理が必要です。修正マイナー則で行う場合は、この処理は行わずに step2 に進んで下さい。

※ 下限値をテーブルで定義された最小 S 値以下とした場合

```
テーブルの最小 S 値(index0 の値)とセル中央値を比較関数を使用して論理値を求めます、
$10 = GTE($6,$2(0)) /*セル中央値の下限値を越えた論理値を求めます*/
```

※ 下限値を指定した場合

説明の都合上、下限値を 10 としています。

```
$10 = GTE($6,10) /*セル中央値の下限値を越えた論理値を求めます*/
```

求めた論理値をデータ抽出関数(ZSP 関数)を使用してセル中央値数列及び計数値数列を再構成します。

```
$6 = ZSP($10,$6) /*セル中央値を論理"1"のみで再構成する*/
$5 = ZSP($10,$5) /*計数値を論理"1"のみで再構成する*/
```

step2 : 被害度演算を行う

被害演算は、補間関数(ITP 関数)を使用して、テーブルからセル中央値に対応した N 値を求め、頻度解析結果の計数値 n を N で割り算して求めます。

case 1 : テーブル軸を S 軸 Log、N 軸 Log として行う場合

```
$7 = 10^ITP(LGT($2),LGT($3),LGT($6)) /* $7 にテーブルを使用して s に対応した N 値を求めます*/
$8 "セル毎の被害度:" = $5/$7 /* $8 は n/N となります*/
$9 "合計被害度:" = SUM($8) /* $9 は Σ(n/N)となります*/
```

case2 : テーブル軸を S 軸 Linear、N 軸 Log として行う場合

```
$7 = 10^ITP($2,LGT($3),$6) /* $7 にテーブルを使用して s に対応した N 値を求める*/
$8 “セル毎の被害度” = $5/$7 /* $8 は n/N となります*/
$9 “合計被害度” = SUM($8) /* $9 は Σ(n/N) となります*/
```

case3: テーブル軸を S 軸 Log、N 軸 Linear として行う場合

```
$7 =ITP(LGT($2),$3,LGT($6)) /* $7 にテーブルを使用して s に対応した N 値を求める*/
$8 “セル毎の被害度” = $5/$7 /* $8 は n/N となります*/
$9 “合計被害度” = SUM($8) /* $9 は Σ(n/N) となります*/
```

case4: テーブル軸を S 軸 Linear、N 軸 Linear として行う場合

```
$7 =ITP($2,$3,$6) /* $7 にテーブルを使用して s に対応した N 値を求める*/
$8 “セル毎の被害度” = $5/$7 /* $8 は n/N となります*/
$9 “合計被害度” = SUM($8) /* $9 は Σ(n/N) となります*/
```

<S-N 曲線を演算式で定義した場合>

step1: S-N 曲線の式を N を求める式として記述します。

例えば、 $S = -3.04 \cdot \ln(N) + 116.5$  の場合、変形して  $N = \text{Exp}((S - 116.5) / -3.04)$  として記述します。

```
$7 = EXP(($6-116.5)/-3.04) /* 演算式を使用してセル中央値に対応した N 値を求めます*/
$8 “セル毎の被害度” = $5/$7 /* $8 は n/N となります*/
$9 “合計被害度” = SUM($8) /* $9 は Σ(n/N) となります*/
```

<被害度演算の事前処理>

① セル中央値及びセル番号を求める

頻度解析 Library では、セルの中央値及びセル番号は扱いません。求める場合は、Script Subroutine の SS0501.prc (cell\_number) を使用します。

説明の都合上、\$20 にセルサイズ、\$21 にセル個数、\$22 に領域フラグが格納されているとします。

```
def ch_name $23 “セル番号”
def ch_name $6 “セル中央値”
call proc cell_number $20,$21,$22,$23,$6 /*セル番号数列とセル中央値数列を生成*/
```

※ 結果の格納先となる\$23 及び\$6 以外は、即値で与えても問題ありません。

※ \$22 の領域フラグは、0=絶対値型、1=正負型 を意味し、使用した頻度解析手法により異なります。

絶対値型の頻度解析手法

1. レインフロー法
2. 極大値・極小法(絶対値)
3. 極大値・極小値法(正領域)
4. 極大値・極小値法(負領域)

5. 振幅法
6. 最大値法
7. 最小値法

正負型の頻度解析手法

1. 極大値・極小法
2. 極大値法
3. 極小値法
4. 最大値・最小値法
5. レベルクロス法
6. レベルクロス法(正傾斜)
7. レベルクロス法(負傾斜)

8. 時間率法

② 編度解析結果が正負領域を絶対値化する

正負領域の場合、セル個数は必ず偶数個で構成されていますので、計数値は半分から折り返す処理を行うことで絶対値化出来ます。

```
$23 = ERC(LEN($23)/2,LEN($23)-1,$23) /*セル番号を正領域のみに再構成します*/
$6 = ERC(LEN($6)/2,LEN($6)-1,$6) /*セル地中央値を正領域のみに再構成します*/
$5 “計数値” = REV(ERC(0,LEN($5)/2-1,$5))+ERC(LEN(LEN($5)/2,LEN($5)-1,$5) /*正負を絶対値化します*/
```

③ 頻度解析結果を基準値に校正する

頻度解析結果を基準時間或いは基準走行距離に換算する必要がある場合は、頻度解析範囲の実計測時間又は実走行距離力と基準値の倍率で計数値を修飾します。\$24 に実計測時間又は実走行距離、\$25 に基準時間又は基準走行距離が格納されているとします。

```
$5 = $5*$25/$24 /* 倍率で$5 の計数値を修飾します*/
```

④ 無効計数値を除去処理する

セルの昇順に、一旦、計数値が 0 以上のセルが出現した後、計数値が 0 のセルが連続した場合に、それ以上の計数値を除去する必要がある場合に行います。

説明の都合上、0 のセルが連続 5 個以上の場合、それ以上のセルの計数値を無視（計数値を 0 とする）ものとし、無効区間の開始 Index 値の取得は、閾値通過持続データ番号関数 2(DTE 関数)を使用します。

```
$26 = DTE(0,0,5,$5-1)          /* $5 の 0 が連続する 5 以上の区間を検索*/  
case $26(0) > 0                /* 連続 5 以上の 0 を検出*/  
  proc modf{  
    $5 = REP(0,ERC(0,$26,$5),$5*0) /*有効計数部分で再構成*/  
  }modf
```